
Приводится реализация решения полной проблемы собственных значений
классическим методом вращений Якоби с использованием функций рабочего
листа MS Excel. 

Целью данной работы является демонстрация возможностей электронных таб-
лиц MS EXСEL для получения способа детальной проверки выполнения заданий
студентами на примере задачи нахождения собственных чисел и собственных векто-
ров классическим методом Якоби [1]. Рассмотрим решение на примере вещественной
симметрической матрицы порядка 3.  

Пусть исходная матрица А располагается на рабочем листе в диапазоне G2:I4 
(рис. 1). Точность вычислений задается в ячейке F1.  

Алгоритм метода состоит в выполнении ряда шагов: 
1) поиск максимального  по модулю внедиагонального элемента;  
2) определение плоскости поворота ( ) 31 ,, ≤≠≤ qpqp ; 
3) определение угла вращения ϕ ; 

4) вычисление матрицы вращений U в плоскости ( )qp,  и матрицы TU ; 

5) получение матрицы ),...;2,1(, )0()1()( AAkUAUA kTk === − , элемент 
)(k 

pqa кото-
рой равен нулю; 
6) определение момента завершения вычислений. 

Рис. 1 

 

 

Рис. 2 

 

Рис. 3 

 

Рис. 4 

Для отбора максимального внедиагонального элемента формируется матрица в 
диапазоне M2:O4 с использованием формулы (1). Формула вводится в ячейку М2 и 
распространяется на указанный диапазон (рис. 2). 

=ЕСЛИ($B2=G$1;0;G2) (1)

{=МАКС(ABS(M2:O4))} (2)

Максимальный по модулю элемент в диапазоне M2:O4 формируется с исполь-
зованием формулы обработки массива (2) в ячейке J2 (рис. 3). 

Плоскость поворота ),( qp  определяется в диапазонах V2:X4, AB2:AB4, 
AC2:AC4 (3)-(5), Y2:AA4, AD(2):AD4, AE2:AE4 (6)-(8)  и окончательно в ячейках 
AI2 и AJ2 (9)-(10) (рис. 4). 

=ЕСЛИ(И(ОКРУГЛ(ABS(G2);6)=ОКРУГЛ($J2;6);$B2<G$1);$F2;"") (3)

=МИН(V2:X2) (4)

=ЕСЛИ(AB2<>0;AB2;100) (5)

=ЕСЛИ(V2<>"";G$1;"") (6)

=МИН(Y2:AA2) (7)

=ЕСЛИ(AD2<>0;AD2;100) (8)

=МИН(AC2:AC4) (9)

=ВПР(AI2;AC2:AE4;3;ЛОЖЬ) (10)
Угол поворота ϕ  вычисляется в ячейке L2 по формуле (11) (рис. 3). 

=ЕСЛИ(ЕОШИБКА(0,5*ATAN(2*СМЕЩ(G2;AI2-1;AJ2-1;1;1)/ 
(СМЕЩ(G2;AJ2-1;AJ2-1;1;1)-СМЕЩ(G2;AI2-1;AI2-1;1;1)))); ЕС-

ЛИ(K2<0;ПИ()/4;-ПИ()/4);0,5*ATAN(2*СМЕЩ(G2;AI2-1;AJ2-1;1;1)/ 
(СМЕЩ(G2;AJ2-1;AJ2-1;1;1)-СМЕЩ(G2;AI2-1;AI2-1;1;1)))) 

(11)

Матрица поворота U  формируется в диапазоне P2:R4 с использованием фор-
мулы 12 (рис. 5). Транспонированная матрица TU (13) располагается в диапазоне 
С2:Е4 (рис. 6). 

=ЕСЛИ((AF$1=$AI2)*($F2=$AI2)+(AF$1=$AJ2)*($F2=$AJ2)>=1;COS($L2); 
ЕСЛИ((AF$1=$AI2)*($F2=$AJ2)=1;-SIN($L2); 
ЕСЛИ((AF$1=$AJ2)*($F2=$AI2)=1;SIN($L2); 

ЕСЛИ((M$1=$F2)*($F2<>$AI2)*($F2<>$AJ2)=1;1;0)))) 

(12)

{=ТРАНСП(P2:R4)} (13)

{=МУМНОЖ(C2:E4;G2:I4)} (14)

{=МУМНОЖ(S2:U4;P2:R4)} (15)

Произведение матриц UAUA kTk )1()( −=  получаем последовательным перемно-
жением )1( −kT AU  (14) и UAU kT )( )1( −  (15) в диапазонах S2:U4 и G5:I7 соответственно 
(рис. 7– 8). 

В ячейке АК2 вычисляется квадратный корень из суммы квадратов внедиаго-
нальных элементов (16). Это значение используется для определения момента завер-
шения итерационного процесса.  

В диапазоне AL2:AL4 формируется вектор собственных чисел (17). На первой 
итерации в диапазон AM2:AO4 помещена матрица U  (18). 

=КОРЕНЬ(СУММКВ(M2:O4)) (16)

=ГПР(F2;$G$1:I4;СТРОКА(F2)) (17)

{=P2:R4} (18)

{=МУМНОЖ(AM2:AO4;P5:R7)} (19)

=ЕСЛИ(AK5<=$F$1;0;A2+1) (20)
Для выполнения второй итерации копируем диапазон В2:Е4 в В5:Е7, J2:AO4 – в 

J5:AL7. В диапазон AM5:AO7 вводим формулу (19) для формирования собственных 
векторов. В ячейку вводится А5 формула (20) для определения момента завершения 
итерационного процесса. 

Последующие итерации получаются копированием диапазона A5:AO7 и после-
дующие вставкой его в диапазоны A8:AO10, A11:AO13 и т. д. до тех пор, пока в 

столбце А не появится нулевое значение, являющееся признаком завершения работы 
алгоритма (рис. 9).  

На рисунке 9 собственные числа и собственные векторы располагаются в диа-
пазонах AL11:AL13 и AM11:AО13 соответственно. Проверка AX X= λ  полученных 
собственных значений и собственных векторов осуществляется в диапазонах  
P17:R19 и P21:R23 (рис. 10−11).  

  

Рис. 5    Рис. 6 
 

  
 

Рис. 7     Рис. 8 
 

 

 

Рис. 9 

Рис. 10  

 
 

Рис. 11 

 
 

Рис. 12 
 

Для быстрой проверки большого числа вариантов можно воспользоваться сле-
дующим приемом. Помещаем в ячейку D1 номер нужного варианта (рис. 13). Зада-
ния для самостоятельного решения располагаются в столбцах AR:AU, причем в 
столбце AR последняя цифра означает номер строки матрицы, а остальные – номер 
варианта. Функция (21) переносит нужную матрицу в область решений. 

=ВПР($D$1*10+$F2;$AR$2:$AU$217;G$1+1) (21)

Нетрудно заметить, что решение задачи с другой размерностью матрицы осу-
ществляется аналогично. 

